


Beispielprogramme für Lexium MDRIVE Motion Control (RS485)

Im Folgenden werden 4 ASCII-Beispielprogramme mit Funktion und Programm-Syntax kurz beschrieben, um die umfangreiche Funktionalität und die einfache Programmierung der
→ **Lexium MDRIVE Motion Control** – Schrittmotorantriebe
mit RS485-Schnittstelle oder Ethernet und digitalen Ein-/Ausgängen (IOs) zu demonstrieren.

Die servoähnliche Closed-Loop-Regelung **HMT (Hybrid-Motion-Technology)** regelt mit Hilfe des integrierten Encoders mit 1000 Linien (4000 Inkremente / Umdrehung) die Drehzahl gegebenenfalls bis zum Stillstand herunter, wenn der Motor über sein Kippmoment hinaus belastet wird. Der Motor fällt nicht mehr „außer Tritt“ und verhält sich wie ein DC-Servo-Antrieb. Über zusätzliche Befehle im ASCII-Befehlssatz können diese Funktionen einfach aktiviert und parametrisiert werden. Servo-typisches Regler-Tuning entfällt. Im Torque-Mode kann eine Drehmomentbegrenzung eingestellt werden, um zum Beispiel Anzugsmomente zu begrenzen.

Die auskommentierten Programme sind farbig so dargestellt, wie diese im Editor-Fenster des SEM-Terminal-Programmes mit der automatischen Befehls- und Syntaxerkennung erscheinen.



Im Editor-Fenster werden Kommentare nach dem Hochkomma grün dargestellt und beim Download ignoriert, wenn pro Zeile nicht mehr als insgesamt 64 Zeichen verwendet werden !

Vor dem Download empfiehlt es sich mit dem Befehl „fd“ die jeweilige Steuerung auf Werkseinstellungen zurückzusetzen und damit auch alte Programme zu löschen.

Folgende Beispielprogramme:

1. Referenzfahrt auf Sensor, Absolut- und Relativpositionierung, selbststartend
2. Positionierung im Hybrid-Mode 0 (aus) und 2 (ein), mit und ohne Nachholen von Schritten
3. Drehen entsprechend Analogeingang, Triggern/Setzen eines Ausgangs nach jeder 1/4 Umdr.
4. Positionierung 0 bis 360° über Eingänge 2 bis 4, Fehlerbehandlung

Für eine einwandfreie Funktion wird ein ordnungsgemäßer Anschluss und Betrieb der jeweiligen Antriebe und Steuerungen gemäß den Hard- und Software-Dokumentationen vorausgesetzt, die unter www.kocomotion.de auf den jeweiligen Produktseiten zum Download zur Verfügung stehen.

1. Referenzfahrt auf Sensor, Absolut- und Relativpositionierung, selbststartend

Der Lexium MDrive-Antrieb mit integriertem Encoder (Closed Loop) arbeitet im Hybrid-Modus mit variablem Motorstrom (AS=2). Eventuell durch Überlast nicht ausgeführte Schritte werden nicht sofort mit überhöhter Drehzahl nachgeholt (MU=0)

Eingang 2 wird als low-aktiver Referenz-Eingang (Home) definiert, so dass ein einfacher Schließer nach Masse (GND) genügt, wenn am Pin „Input Reference“ eine Logik-Spannung anliegt und damit npn-Verhalten (sinking) erzeugt wird.

Ausgang 3 wird als low-aktiver Fehler-Ausgang (Error) definiert, so dass dieser immer dann aktiv ist, wenn ein Fehler auftritt. Dies passiert ist z.B. bei kurzzeitiger Überlastung des Motors, wenn die Drehzahl zwischenzeitlich soweit herunterregelt wird, dass die mit dem Befehl Lg=5000 „Lag Limit“ definierte Wegdifferenz in Mikroschritten (siehe unten – 0,1 Umdr.) überschritten wird. Diese fehlenden Mikroschritte werden dann im Bewegungsprofil angehängt, so dass die Positionierung trotz zwischenzeitlicher Überlast korrekt ausgeführt wird.

Die LED 2 wird mit dem AO-Befehl (Attention Output Mask) so definiert, dass diese von Grün auf Rot schaltet sobald die Closed-Loop-Regelung HMT im Überlastfall eingreift und die Motordrehzahl regelt.

Das Programm führt eine Referenzfahrt auf Eingang 2 aus, setzt den Positionszähler auf Null und führt dann eine Absolut-Positionierung von genau 1 Umdrehung aus und fährt dann relativ 0,2 Umdrehungen zurück. Vor und nach den Positionierungen wird der Wert des Positionszählers auf die Schnittstelle RS485 ausgegeben und erscheint z.B. im Terminal-Fenster des IMS-Terminal-Programmes.

Das Programm wird gestartet durch die Eingabe des Befehls „EX su“. Der Name „su“ bedeutet „Startup“ und ist der Autostart-Name, so dass das Programm ab dieser Adresse auch ohne Start-Befehl und nur nach dem Zuschalten der jeweiligen Versorgungsspannung startet.

'Einstellungen

```
AS=2          'Closed-Loop-Regelung (HMT) 2: var. Motorstrom
MS=250       '250 Mikroschr., d.h. 50.000 MS pro Umdr.
Vi=5000      'Startdrehz. 5.000 MS/s = 0,1 Umdr./s
Vm=200000    'Enddrehz. 200.000 MS/s = 4 Umdr./s
A=400000     'Beschl. 400.000 MS/s² = 8 Umdr./s²
D=A          'Bremsrampe = Beschleunigung
HC=5         'Haltestrom 5%
RC=40        'Laufstrom 40%
Cb=1         'Control bounds: 1,3 Vollschr.
Mu=0         'Makeup mode 0 - kein sofortiges Nachholen
Ld=5000      'lead limit: 5.000 MS = 0,1 Umdr.
Lg=5000      'lag limit: 5.000 MS = 0,1 Umdr.
Lt=1000      'Locked rotor timeout 1000 ms

Is=2,1,0     'Eingang 2 als Home-Eingang, low-aktiv
Os=3,18,0    'Ausgang 3 als Fehler-Ausgang, low-aktiv
Ao=16        'LED 2 an wenn Regelung eingreift (HMT active)
```

'Programm

```
PG 1         'Start Programm-Mode Adresse 1
LB Su        'Programmname "su" - Startup
HM 1         'Referenzfahrt auf Home-Schalter
H            'warten bis Beweg. beendet
P=0          'Positionszähler auf 0 setzen
PR "P=",P    'Anzeige Position
MA 50000     'Absolutbew. 50.000 Schritte = 1 Umdr.
H            'warten bis Beweg. beendet
PR "P=",P    'Anzeige Position
MR -10000    'Relativbew. -10.000 Schritte = -0,2 Umdr.
H            'warten bis Beweg. beendet
PR "P=",P    'Anzeige Position
E            'Ende

PG           'Beenden Programm-mode
S            'Abspeichern nach Download
```

2. Positionierung im Hybrid-Mode 0 (aus) und 2 (ein), mit und ohne Nachholen von Schritten

Es werden jeweils 10 Umdrehungen ausgeführt in 3 Betriebsarten der Hybrid-Regelung:

- AS=0: Hybrid-Regelung ausgeschaltet, Motor fällt bei Überlast außer Tritt und bleibt stehen
- AS=2, MU=0: Hybrid-Regelung eingeschaltet, variabler Motorstrom, Drehzahl wird bei Überlast heruntergeregelt, kein Nachholen der nicht ausgeführten Schritte mit überhöhter Drehzahl, kein Nachregeln der Position nach Beendigung der Bewegung
- AS=2, MU=1: Hybrid-Regelung eingeschaltet, variabler Motorstrom, Drehzahl wird bei Überlast heruntergeregelt, aktive Nachregelung (Make-Up) eingeschaltet: nicht ausgeführte Schritte werden während der Bewegung mit überhöhter Drehzahl nachgeholt, Nachregeln der Position am Bewegungsende mit definierter Drehzahl (MF)

Die Ausgänge 1 bis 3 sind als Status-Ausgänge definiert.

Über die Interrupt-Funktion „OE“ (Error-Handler) wird eine Fehlerbehandlungs-Routine aktiviert, die im Fehlerfall ausgeführt wird und entsprechende Fehler über die RS485 meldet und dann zurücksetzt.

Das Programm wird gestartet durch den Befehl „EX xx“ oder durch Schalten des Eingangs 1 (GO-Eingang).

'Einstellungen

```
AS=2      'Hybrid-Mode ein, variabler Motorstrom
Cb=1      'Control bounds: 1,3 Vollschritte
Hc=20     'Haltestrom 20%
Rc=40     'Laufstrom 40%
Ms=250    '250 Mikroschr., d.h. 50.000 MS pro Umdr.
Vi=5000   'Startdrehz. 5.000 MS/s = 0,1 Umdr./s
Vm=200000 'Enddrehz. 200.000 MS/s = 4 Umdr./s
A=400000  'Beschl. 400.000 MS/s² = 8 Umdr./s²
D=A       'Bremsrampe = Beschleunigung
Ld=10000  'lead limit: 10.000 MS = 0,2 Umdr.
Lg=10000  'lag limit: 10.000 MS = 0,2 Umdr.
Lt=1000   'Locked rotor timeout 1000 ms
Mf=100000 'Makeup-frequ. 100.000 MS/s = 2 Umdr./s

Is=1,4,0  'Eingang 1 als GO/Start-Eingang, low-aktiv
Os=1,18,0 'Ausgang 1 als Fehler-Ausgang, low-aktiv
Os=2,17,0 'Ausgang 2 als Moving-Ausgang, low-aktiv
Os=3,25,0 'Ausgang 3 als Makeup-Ausgang, low-aktiv
Ao=16     'LED 2 an wenn Regelung eingreift (HMT active)
```

'Programm

```
PG 1      'Start Programm-Mode Adresse 1
LB Xx     'Programmname "Xx"
OE E1     'Unterprogramm Error-Handler "E1" aktiviert
P=0       'Positionszähler auf 0 setzen
PR "10 Umdr. Hybridmode aus"
  As=0    'Hybrid-Mode aus, kein Nachregeln !
  H 1000  '1 s warten
  MR 500000 'Relativbew. 500.000 Schritte = 10 Umdr.
  H       'warten bis Beweg. beendet
PR "10 Umdr. Hybridmode variabler Motorstrom"
  As=2    'Hybrid-Mode ein, variabler Motorstrom
  Mu=0    'Makeup/Nachholen der Schritte ausgeschaltet
  H 1000  '1 s warten
  MR 500000 'Relativbew. 500.000 Schritte = 10 Umdr.
  H       'warten bis Beweg. beendet
PR "10 Umdr. Hybridmode variabler Motorstrom, MU aktiv"
  As=2    'Hybrid-Mode ein, variabler Motorstrom
  Mu=1    'Makeup/Nachholen aktiv mit MF am Ende
  H 1000  '1 s warten
  MR 500000 'Relativbew. 500.000 Schritte = 10 Umdr.
  H       'warten bis Beweg. beendet
E         'Ende

LB E1     'Unterprogramm "e1" Error-Handler
PR "Er=",Er 'Anzeige Error/Fehler
Er=0     'Fehler-Variablen auf 0 zurücksetzen
H 100    '100 ms warten
CF       'Clear locked rotor fault - falls vorhanden
RT       'Rücksprung aus Unterprogramm

PG       'Beenden Programm-mode
S        'Abspeichern nach Download
```

3. Drehzahl entsprechend Analogeingang, Triggern/Setzen eines Ausgangs nach jeder 1/4 Umdr.

Der Eingang 1 wird als START-Eingang, der Eingang 2 als STOP-Eingang definiert. Beide sind als low aktive Eingänge definiert, so dass ein einfacher Schließer nach Masse (GND) genügt, wenn am Pin „Input Reference“ eine Logik-Spannung anliegt und damit npn-Verhalten (sinking) erzeugt wird.

Das Programm arbeitet in einer Endlosschleife und fragt den Analog-Eingang ab und lässt den Motor entsprechend des Wertes am Analogeingang mit unterschiedlichen Drehzahlen drehen. Die maximale Drehzahl beträgt ca. 4 Umdr./s (Analogwert max. $4096 \times 50 = 204800$).

Ausgang 3 ist als Status-Ausgang „Moving“ definiert und wird immer dann geschaltet, wenn der Motor in Bewegung ist.

Die LED 2 wird mit dem AO-Befehl (Attention Output Mask) so definiert, dass diese von Grün auf Rot schaltet sobald die Closed-Loop-Regelung HMT im Überlastfall eingreift und die Motordrehzahl regelt.

Der ASCII-Befehlssatz MCODE ermöglicht mit den sogenannten TRIP-Befehlen auch schnelle ereignisgesteuerte Interrupt-Funktionen.

Im Programm wird über die HighSpeed Interrupt-Funktion „TR“ Trip Relative auf Positionen getriggert. Der schnelle Ausgang 3 wird hier so angesteuert, dass nach jeder 1/4 Umdrehung (alle 12500 Mikroschritte) der Ausgang für $10\mu\text{s}$ ($200 \times 50\text{ns}$) gesetzt wird, um z.B. eine Kamera zu triggern oder Ähnliches. Dies wird exakt nach jeder 1/4 Umdrehung ausgeführt, zeitlich abhängig von der Drehzahl oder dem Eingreifen des Reglers bei Überlast !

Das Programm wird gestartet durch den Befehl „EX xx“ oder durch Schalten des Eingangs 1 (GO-Eingang).

```
'Einstellungen
AS=2          'Closed-Loop-Regelung (HMT) 2: var. Motorstrom
MS=250       '250 Mikroschr., also 50.000 MS/Umdr.
VI=5000      'Startdrehzahl 0,1 Umdr./s
VM=250000    'max. Drehzahl 5 Umdr./s
A=1000000    'Beschleunigung 20 Umdr./s²
D=A          'Bremsrampe = Beschleunigung
HC=10        'Haltestrom 10%
RC=50        'Laufstrom 50%
LD=5000      'Lead limit 5.000 MS = 0,1 Umdr.
LG=5000      'Lag limit 5.000 MS = 0,1 Umdr.
MU=0         'Makeup mode 0 - kein sofortiges Nachholen

IS=1,4,0     'Eingang 1 als START Eingang, low-aktiv
IS=2,5,0     'Eingang 2 als STOP Eingang, low-aktiv
IS=5,9,0     'Analog-Eing., Spannung 0...5V

OS=1,17,0    'Ausgang 1 als Moving-Ausgang, low-aktiv
OS=3,28,0    'Ausgang 3 als TRIP-Ausgang, low-aktiv
AO=16        'LED 2 an wenn Regelung eingreift (HMT active)
ER=0

'Programm
PG 1         'Programmiermodus ab Adresse 1
LB XX       'Programmname / Label XX
SL 0        'Anhalten falls vorher Bewegung ablief
H
C1=0        'Positionszaehler/Motorzaehler auf 0
C2=0        'Encoderzaehler auf 0
TR=12500,0,0 'TR definiert (nach 0,25 Umdr.)
CW=200      '200 x 50ns Pulslänge Trip-Ausg.
TE=16       'TR aktiviert
LB X1       'Label X1
R1=I5       'Analogwert in Register R1
R1=R1*50    'R1 mit 50 multiplizieren
SL R1       'Drehen mit errechnetem Wert in MS/s
H 100      '100ms warten
BR X1       'Sprung zu X1 - Endlosschleife
E          'Programmende

PG         'Ende Programmiermodus
S         'Abspeichern nach Download
```

4. Positionierung 0 bis 360° über Eingänge 2 bis 4, Fehlerbehandlung

Die Eingänge 2, 3 und 4 werden als normale low aktive Eingänge definiert, so dass ein einfacher Schließer nach Masse (GND) genügt, wenn am Pin „Input Reference“ eine Logik-Spannung anliegt und damit npn-Verhalten (sinking) erzeugt wird.

Das Programm arbeitet in einer Endlosschleife und fragt den Zustand der Eingänge 2 bis 4 ab und positioniert den Motor in Abhängigkeit davon absolut auf 0°, 90°, 180°, 270° oder 360°. Dabei werden die Eingänge 2 bis 4 als 3-bit-Wort eingelesen und dann entsprechende Unterprogramme aufgerufen.

Ausgang 3 ist als Status-Ausgang „Moving“ definiert und wird immer dann geschaltet, wenn der Motor in Bewegung ist.

Über die Interrupt-Funktion „OE“ (Error-Handler) wird eine Fehlerbehandlungs-Routine aktiviert, die im Fehlerfall ausgeführt wird. Dabei wird der entsprechende Fehler über die RS485 gemeldet und nach 1 Sekunde zurückgesetzt.

Das Programm wird gestartet durch die Eingabe des Befehls „EX su“. Der Name „su“ bedeutet „Startup“ und ist der Autostart-Name, so dass das Programm ab dieser Adresse auch ohne Start-Befehl und nur nach dem Zuschalten der jeweiligen Versorgungsspannung startet. und damit ein Überlastfall auftritt.

'Einstellungen

```
AS=2      'Closed-Loop-Regelung (HMT) 2: var. Motorstrom
MS=250    '250 Mikroschr., also 50.000 MS/Umdr.
Vi=5000   'Startdrehzahl 0,1 Umdr./s
Vm=250000 'max. Drehzahl 5 Umdr./s
A=1000000 'Beschleunigung 20 Umdr./s²
D=A       'Bremsrampe = Beschleunigung
Hc=10     'Haltestrom 10%
Rc=50     'Laufstrom 50%
Ld=5000   'Lead limit 5.000 MS = 0,1 Umdr.
Lg=5000   'Lag limit 5.000 MS = 0,1 Umdr.
Mu=0      'Makeup mode 0 - kein sofortiges Nachholen

Is=2,0,0  'Eingang 2 als allg. Eingang, low-aktiv
Is=3,0,0  'Eingang 3 als allg. Eingang, low-aktiv
Is=4,0,0  'Eingang 4 als allg. Eingang, low-aktiv
Os=3,17,0 'Ausgang 3 als Moving-Ausgang, low-aktiv
Ao=16     'LED 2 an wenn Regelung eingreift (HMT active)
```

'Programm

```
PG 1      'Start Programm-Modus ab Adresse 1
LB Su     'Programmname su (startup) - Autom. Start
OE Yy     'Error-Handler "yy" aktivieren
P=0       'Positionszähler auf Null
LB Y1     'Marke Y1 - Endlosschleife
R1=In     'Einlesen I2 bis I4 als 4 Bit
R1=R1&14  'bitweise Vergleich mit 14 (I1 ignorieren)
CL X1,R1=14 'Aufruf x1 -> 0°      I1/2/3/4: -,0,0,0
CL X2,R1=12 'Aufruf x2 -> 90°     I1/2/3/4: -,1,0,0
CL X3,R1=10 'Aufruf x3 -> 180°    I1/2/3/4: -,0,1,0
CL X4,R1=8  'Aufruf x4 -> 270°   I1/2/3/4: -,1,1,0
CL X5,R1=6  'Aufruf x5 -> 360°   I1/2/3/4: -,0,0,1
H 10
BR Y1     'Endlosschleife
E         'Ende Hauptprogramm

LB X1     'Unterprogramm x1 -> 0°
MA 0      'absolut auf 0° fahren
H         'warten bis Bew. beendet
RT       'Rücksprung aus Unterprogramm

LB X2     'Unterprogramm x2 -> 90°
MA 12500  'absolut auf 90° fahren
H         'warten bis Bew. beendet
RT       'Rücksprung aus Unterprogramm
```

```

LB X3      'Unterprogramm x3 -> 180°
MA 25000  'absolut auf 180° fahren
H          'warten bis Bew. beendet
RT        'Rücksprung aus Unterprogramm

LB X4      'Unterprogramm x4 -> 270°
MA 37500  'absolut auf 270° fahren
H          'warten bis Bew. beendet
RT        'Rücksprung aus Unterprogramm

LB X5      'Unterprogramm x5 -> 360°
MA 50000  'absolut auf 360° fahren
H          'warten bis Bew. beendet
RT        'Rücksprung aus Unterprogramm

LB Yy      'Unterprogramm Error-Handler
PR "Fehler=",Er
H 100     '100 ms warten für PR-Befehl
Er=0      'Fehler-Variable zurücksetzen
H 1000    '1 s warten
CF        'wenn nötig Locked-Rotor zurücksetzen
RT        'Rücksprung aus Unterprogramm

E          'Programm-Ende

PG        'Ende Programm-Modus
S         'Abspeichern im EEPROM

```

Änderungen vorbehalten / Rev. 07/2016



KOCO MOTION GmbH
Niedereschacher Straße 54
78083 Dauchingen
Germany
Telefon +49 7720 995858-0
Telefax +49 7720 995858-99
info@kocomotion.de
www.kocomotion.de

Ihre technischen Berater im Innen- und Außendienst:

Olaf Kämmerling
Verkaufsleitung
Telefon +49 7720 995858-13
Telefax +49 7720 995858-99
o.kaemmerling@kocomotion.de

Frank Freund
Innendienst
Telefon +49 7720 995858-17
Telefax +49 7720 995858-99
f.freund@kocomotion.de

Helmut Rentergent
Außendienst und technische
Beratung DC-Motoren
Telefon +49 7082 92257
Telefax +49 7082 92259
h.rentergent@kocomotion.de

Holger Ruhland
Außendienst PLZ 0, 1, 2,
38 – 39, 8 – 87, 90 – 96, 98 – 99
Telefon +49 35205 4587-8
Telefax +49 35205 4587-9
h.ruhland@kocomotion.de

Jürgen Kühne
Außendienst
PLZ 30 – 37, 4, 5, 6, 97
Telefon +49 7253 988160
Telefax +49 7253 988159
j.kuehne@kocomotion.de

Neil Engstrom
Außendienst
PLZ 7, 88 + 89
Telefon +49 7720 995858-0
Telefax +49 7720 995858-99
n.engstrom@kocomotion.de